Q:

Points on axis only half-shown

I need to add two points on a Plot. As they're on the axes, and PlotRange->{{0,27},{0,27}} only half the points are shown. I don't want to increase the PlotRange to include negative values, and I'd rather not have the axes extending below (0,0).
Is there any other option? My code is the following:
Edited full code
Plot[
 {Piecewise[{
    {(5 - .5*q^(1/2))^2, q <= 100/9},
    {(10 - 2*q^(1/2))^2, 100/9 < q < 25}}],
  Piecewise[{
    {(10 - 2*q^(1/2))^2, q <= 100/9},
    {(5 - .5*q^(1/2))^2, 100/9 < q < 30}}],
  25 - q}, {q, 0, 30},
 PlotRange -> {{0, 27}, {0, 27}},
 PlotRangePadding -> .5,
 AspectRatio -> 1,
 Exclusions -> None,
 PlotStyle -> {Black, Directive[Black, Dashed], 
   Directive[Black, Thickness[.005]]},
 Filling -> {1 -> Axis},
 BaseStyle -> {FontFamily -> "Times New Roman", 12},
 AxesStyle -> Directive[Black, Thick],
 Ticks -> None,
 AxesLabel -> {HoldForm[Subscript[q, HoldForm[A]]], 
   HoldForm[Subscript[q, HoldForm[B]]]},
 Epilog -> {
   Text[HoldForm[L] == 50, {8, 25}],
   Text[HoldForm[K] == 50, {22, 8}],
   Text[HoldForm[q] == 25, {5, 22}],
   Text[HoldForm[R], {25, 1}],
   Text[HoldForm[S], {1, 25}],
   PointSize -> .025,
   Point[{{25, 0}, {0, 25}}]

}
]

A:

Use ImagePadding and PlotRangeClipping:
Plot[
 {Piecewise[{{(5 - .5*q^(1/2))^2, q <= 100/9}, {(10 - 2*q^(1/2))^2, 
     100/9 < q < 25}}], 
  Piecewise[{{(10 - 2*q^(1/2))^2, q <= 100/9}, {(5 - .5*q^(1/2))^2, 
     100/9 < q < 30}}], 25 - q}, {q, 0, 27},
 PlotRange -> {{0, 27}, {0, 27}},
 AspectRatio -> 1,
 Exclusions -> None,
 AxesLabel -> {"x", "y"},
 Filling -> {1 -> Axis},
 Ticks -> None,
 Epilog -> {PointSize -> .025, Point[{{25, 0}, {0, 25}}]},
 ImagePadding -> 20,
 PlotRangeClipping -> False
 ]

This works by first expanding the white-space around the plot area, and then making sure that objects are not clipped at the edge of the plot range. Note that I have change the range for q to go only to 27, so that the lines end where the axis ends (otherwise, the lines will go over the axis label)

